
Rozdział 4

Programowanie równoległe
w języku CUDA C

W poprzednim rozdziale wykazaliśmy, jak łatwo jest napisać program wykonywany przez GPU.
Obliczyliśmy nawet sumę dwóch liczb, aczkolwiek niezbyt dużych, bo tylko 2 i 7. Przyznajemy,
tamten przykład nie był zbyt porywający, ani też praktyczny. Mamy jednak cichą nadzieję, że
dzięki niemu mogłeś się przekonać, iż pisanie programów w CUDA C to nic trudnego, i że obudzili-
śmy w Tobie ciekawość, aby dowiedzieć się więcej na ten temat. Jedną z największych zalet wy-
konywania obliczeń na procesorze GPU jest możliwość wykorzystania jego potencjału w zakresie
przetwarzania równoległego. Dlatego w tym rozdziale znajduje się opis technik równoległego
wykonywania kodu CUDA C na GPU.

4.1. Streszczenie rozdziału
W tym rozdziale:

� Poznasz podstawową technikę programowania równoległego CUDA.

� Napiszesz pierwszy równoległy program w języku CUDA C.

4.2. Programowanie równoległe w technologii CUDA
W jednym z poprzednich rozdziałów pokazaliśmy, jak spowodować wykonanie standardowej
funkcji języka C na urządzeniu. W tym celu należy do funkcji dodać słowo kluczowe __global__,
a następnie wywołać ją za pomocą specjalnej składni z użyciem nawiasów trójkątnych. Nie
dość, że jest to technika prymitywna, to na dodatek jeszcze i bardzo nieefektywna, gdyż spece
z NVIDII przecież tak zaprojektowali procesory graficzne, aby mogły wykonywać setki obliczeń
równocześnie. Na razie nie skorzystaliśmy z tej możliwości, ponieważ dotychczasowe programy
zawierały tylko jądro działające na GPU szeregowo. W tym rozdziale dowiesz się, jak napisać
jądro wykonujące obliczenia równolegle.

4.2.1. SUMOWANIE WEKTORÓW

Poniżej przedstawiamy prosty program, na którego przykładzie wprowadzimy pojęcie wątków
i pokażemy, jak ich używać. Przypuśćmy, że mamy dwie listy liczb i chcemy zsumować ich ele-
menty znajdujące się na odpowiadających sobie pozycjach, a następnie wyniki zapisać w trzeciej
liście. Ilustracja przebiegu tego procesu znajduje się na rysunku 4.1. Osoby znające algebrę liniową
od razu rozpoznają, że jest to sumowanie dwóch wektorów.

Rysunek 4.1. Sumowanie dwóch wektorów

SUMOWANIE WEKTORÓW PRZY UŻYCIU PROCESORA CPU

Najpierw zobaczymy, jak taką operację można wykonać za pomocą zwykłego kodu w języku C:

#include "../common/book.h"
#define N   10
void add( int *a, int *b, int *c ) {
    int tid = 0;    // To jest CPU nr zero, a wi�c zaczynamy od zera
    while (tid < N) {
        c[tid] = a[tid] + b[tid];
        tid += 1;   // Mamy tylko jeden CPU, a wi�c zwi�kszamy o jeden
    }
}
int main( void ) {
    int a[N], b[N], c[N];
    // Zape�nienie tablic a i b danymi za pomoc� CPU
    for (int i=0; i<N; i++) {
        a[i] = -i;
        b[i] = i * i;
    }
    add( a, b, c );
    // Wy�wietlenie wyników
    for (int i=0; i<N; i++) {
        printf( "%d + %d = %d\n", a[i], b[i], c[i] );
    }
    return 0;
}

Większa część kodu tego programu nie wymaga objaśnień. Napiszemy tylko kilka słów o funkcji
add(), aby wytłumaczyć się z tego, dlaczego ją niepotrzebnie skomplikowaliśmy.

void add( int *a, int *b, int *c ) {
    int tid = 0;    // To jest CPU zero, a wi�c zaczynamy od zera
    while (tid < N) {
        c[tid] = a[tid] + b[tid];
        tid += 1;   // Mamy tylko jeden CPU, a wi�c zwi�kszamy o jeden
    }
}

Suma obliczana jest za pomocą pętli while, w której zmienna indeksowa o nazwie tid przyj-
muje wartości od 0 do N–1. Sumowane są kolejno odpowiadające sobie elementy tablic a[] i b[],
a wyniki są zapisywane w odpowiednich elementach tablicy c[]. Działanie to można by było
zapisać prościej:

void add( int *a, int *b, int *c ) {
    for (i=0; i < N; i++) {
        c[i] = a[i] + b[i];
    }
}

Skorzystaliśmy z nieco bardziej pokrętnej metody, aby uwidocznić możliwość zrównoleglenia
tego kodu, gdyby działał w systemie wieloprocesorowym lub z procesorem wielordzeniowym.
Gdyby na przykład procesor był dwurdzeniowy, to można by było zmienić wartość inkremen-
tacji na 2 i dla pierwszego rdzenia zainicjować pętlę z wartością tid = 0, a dla drugiego z war-
tością tid = 1. Wówczas pierwszy rdzeń sumowałby elementy znajdujące się pod indeksami
parzystymi, a drugi — pod indeksami nieparzystymi. W związku z tym na poszczególnych
rdzeniach procesora byłby wykonywany następujący kod:

RDZEŃ 1

void add( int *a, int *b, int *c )
{
    int tid = 0;
    while (tid < N) {
        c[tid] = a[tid] + b[tid];
        tid += 2;
    }
}

RDZEŃ 2

void add( int *a, int *b, int *c )
{
    int tid = 1;
    while (tid < N) {
        c[tid] = a[tid] + b[tid];
        tid += 2;
    }
}

Oczywiście, aby to zadziałało zgodnie z opisem, trzeba by było napisać sporo dodatkowego kodu.
Należałoby utworzyć wątki robocze do wykonywania funkcji add() oraz przyjąć założenie, że
wszystkie wątki będą działać równolegle, co niestety nie zawsze jest prawdą.

SUMOWANIE WEKTORÓW ZA POMOCĄ PROCESORA GPU

Działanie to można zrealizować w bardzo podobny sposób na procesorze GPU, pisząc funkcję
add() dla urządzenia. Kod będzie podobny do tego, który został już pokazany. Najpierw jednak
zapoznamy się z funkcją main(). Mimo że jej implementacja dla GPU jest nieco inna niż dla
CPU, to nie ma w niej jednak nic nowego:

#include "../common/book.h"
#define N   10
int main( void ) {
    int a[N], b[N], c[N];
    int *dev_a, *dev_b, *dev_c;
    // Alokacja pami�ci na GPU
    HANDLE_ERROR( cudaMalloc( (void**)&dev_a, N * sizeof(int) ) );
    HANDLE_ERROR( cudaMalloc( (void**)&dev_b, N * sizeof(int) ) );
    HANDLE_ERROR( cudaMalloc( (void**)&dev_c, N * sizeof(int) ) );
    // Zape�nienie tablic a i b na CPU
    for (int i=0; i<N; i++) {
        a[i] = -i;
        b[i] = i * i;
    }
    // Kopiowanie tablic a i b do GPU
    HANDLE_ERROR( cudaMemcpy( dev_a, a, N * sizeof(int),
                              cudaMemcpyHostToDevice ) );
    HANDLE_ERROR( cudaMemcpy( dev_b, b, N * sizeof(int),
                              cudaMemcpyHostToDevice ) );
    add<<<N,1>>>( dev_a, dev_b, dev_c );
    // Kopiowanie tablicy c z GPU do CPU
    HANDLE_ERROR( cudaMemcpy( c, dev_c, N * sizeof(int),
                              cudaMemcpyDeviceToHost ) );
    // Wy�wietlenie wyniku
    for (int i=0; i<N; i++) {
        printf( "%d + %d = %d\n", a[i], b[i], c[i] );
    }
    // Zwolnienie pami�ci alokowanej na GPU
    cudaFree( dev_a );
    cudaFree( dev_b );
    cudaFree( dev_c );
    return 0;
}

Można łatwo zauważyć pewne powtarzające się wzorce:

� Alokacja trzech tablic na urządzeniu za pomocą funkcji cudaMalloc(): tablice dev_a i dev_b
zawierają dane wejściowe, a dev_c — wyniki.

� Ponieważ leży nam na sercu czystość środowiska, sprzątamy po sobie za pomocą funkcji
cudaFree().

� Za pomocą funkcji cudaMemcpy() z parametrem cudaMamcpyHostToDevice kopiujemy
dane wejściowe na urządzenie, a następnie kopiujemy wynik do hosta za pomocą tej
samej funkcji z parametrem cudaMemcpyDeviceToHost.

� Uruchamiamy funkcję add() urządzenia w funkcji main() na hoście, używając składni
z trzema nawiasami trójkątnymi.

Przy okazji warto wyjaśnić, dlaczego tablice są zapełniane danymi przez CPU. Nie ma żadnego
konkretnego powodu, aby tak było. Gdyby w dodatku operację tę przeniesiono na GPU, to by
została wykonana szybciej. Jednak celem tego przykładu było zaprezentowanie sposobu im-
plementacji konkretnego algorytmu (w tym przypadku sumowania wektorów) do wykonania
na procesorze GPU. Wyobraź sobie, że jest to tylko jeden z wielu etapów wykonywania jakiejś
większej aplikacji, w której tablice a[] i b[] zostały utworzone przez jakiś inny algorytm albo
wczytane z dysku twardego. Po prostu udawajmy, że dane pojawiły się nie wiadomo skąd i że
trzeba coś z nimi zrobić.

Wracając do sedna, kod źródłowy tej funkcji add() jest podobny do poprzedniej implementacji
dla CPU:

__global__ void add( int *a, int *b, int *c ) {
    int tid = blockIdx.x;    // Dzia�anie na danych znajduj�cych si� pod tym indeksem
    if (tid < N)
        c[tid] = a[tid] + b[tid];
}

I znowu widać znany już wzorzec postępowania:

� Ta funkcja add() zostanie wykonana na urządzeniu. Spowodowaliśmy to poprzez
dodanie do standardowego kodu tej funkcji w języku C słowa kluczowego __global__.

Jak na razie nie pokazaliśmy jeszcze nic nowego, pomijając fakt, że ten program już nie sumuje
liczb 2 i 7. A jednak są dwie rzeczy warte uwagi. Nowe są parametry w nawiasach trójkątnych
oraz kod źródłowy jądra.

Do tej pory funkcja jądra była zawsze wywoływana za pomocą następującej ogólnej składni:

j�dro<<<1,1>>>( param1, param2, … );

Natomiast tym razem zmieniła się liczba w nawiasach:

add<<<N,1>>>( dev _ a, dev _ b, dev _ c );

O co tu chodzi?

Przypomnijmy, że liczby w nawiasach trójkątnych pozostawiliśmy bez objaśnienia. Napisaliśmy
jedynie, że stanowią one dla systemu wykonawczego informację o sposobie uruchomienia jądra.
Pierwsza z nich określa liczbę równoległych bloków, w których urządzenie ma wykonywać jądro.
W tym przypadku została podana wartość N.

Gdyby na przykład w programie użyto wywołania jądra kernel<<<2,1>>>(), to system wyko-
nawczy utworzyłby dwie jego kopie i wykonywałby je równolegle. Każde z takich równoległych
wywołań nazywa się blokiem. Gdyby napisano wywołanie kernel<<<256,1>>>(), to system
utworzyłby 256 bloków wykonywanych równolegle na GPU. Programowanie równoległe jeszcze
nigdy nie było takie proste.

Teraz nasuwa się pytanie: skoro GPU wykonuje N kopii funkcji jądra, to jak poznać, który blok
wykonuje daną kopię kodu? Aby odpowiedzieć na to pytanie, musimy poznać drugą z nowości
wprowadzonych w tej aplikacji. Znajduje się ona w kodzie jądra, a konkretnie chodzi o zmienną
blockIdx.x:

__global__ void add( int *a, int *b, int *c ) {
    int tid = blockIdx.x;    // Dzia�anie na danych znajduj�cych si� pod tym indeksem
    if (tid < N)
        c[tid] = a[tid] + b[tid];
}

Na pierwszy rzut oka wydaje się, że zmienna ta powinna podczas kompilacji spowodować błąd
składni, ponieważ przypisujemy ją do zmiennej tid, mimo że nigdzie nie ma jej definicji.
A jednak zmiennej blockIdx nie trzeba definiować, ponieważ jest to jedna ze standardowych
zmiennych systemu wykonawczego CUDA. Jej przeznaczenia można domyślić się po nazwie,
a najciekawsze jest to, że używamy jej nawet zgodnie z przeznaczeniem. Zawiera ona indeks
bloku, który aktualnie wykonuje dany kod urządzenia.

Dlaczego w takim razie zmienna ta nie nazywa się po prostu blockIdx, tylko blockIdx.x?
Ponieważ w języku CUDA C można definiować grupy bloków w dwóch wymiarach. Jest to
przydatne w rozwiązywaniu dwuwymiarowych problemów, np. wykonywaniu działań na ma-
cierzach albo przy przetwarzaniu grafiki, gdyż pozwala uniknąć kłopotliwego zamieniania
współrzędnych liniowych na prostokątne. Nie masz się co przejmować, jeśli nie wiesz, o co chodzi.
Po prostu pamiętaj, że czasami indeksowanie dwuwymiarowe jest wygodniejsze od jednowy-
miarowego. Ale nie musisz z tego korzystać. Nie pogniewamy się.

Liczbę równoległych bloków w wywołaniu jądra ustawiliśmy na N. Zbiór równoległych bloków
nazywa się siatką. Zatem nasze wywołanie informuje system wykonawczy, że chcemy utworzyć
jednowymiarową siatkę zawierającą N bloków (wartości skalarne są interpretowane jako jed-
nowymiarowe). Każdy z tych wątków będzie miał inną wartość zmiennej blockIdx.x, a więc
pierwszy będzie miał 0, a ostatni N–1. Wyobraź sobie cztery bloki, wszystkie wykonujące ten sam
kod urządzenia, ale każdy z inną wartością zmiennej blockIdx.x. Poniżej znajduje się kod, jaki
zostałby wykonany przez każdy z tych czterech bloków po podstawieniu w miejsce zmiennej
blockIdx.x odpowiedniej wartości:

BLOK 1

__global__ void
add( int *a, int *b, int *c ) {
    int tid = 0;
    if (tid < N)
        c[tid] = a[tid] + b[tid];
}

BLOK 2

__global__ void
add( int *a, int *b, int *c ) {
    int tid = 1;
    if (tid < N)
        c[tid] = a[tid] + b[tid];
}

BLOK 3

__global__ void
add( int *a, int *b, int *c ) {
    int tid = 2;
    if (tid < N)
        c[tid] = a[tid] + b[tid];
}

BLOK 4

__global__ void
add( int *a, int *b, int *c ) {
    int tid = 3;
    if (tid < N)
        c[tid] = a[tid] + b[tid];
}

Jeśli pamiętasz kod dla CPU pokazany na początku, to pamiętasz też, że w celu obliczenia sumy
wektorów trzeba było przejść przez indeksy od 0 do N–1. Ponieważ system wykonawczy, wy-
wołując blok, od razu wstawia w nim jeden z tych indeksów, wykonuje więc on za nas większość
pracy. A ponieważ nie jesteśmy zbyt pracowici, bardzo nam się to podoba, ponieważ dzięki te-
mu mamy więcej czasu na pisanie na blogu o tym, jak nam się nic nie chce.

A oto ostatnie pytanie, które do tej pory pozostawało bez odpowiedzi: dlaczego sprawdzamy,
czy zmienna tid ma wartość mniejszą od N? Okazuje się, że zmienna ta zawsze powinna być
mniejsza od N, ponieważ tak uruchomiliśmy jądro, iż warunek ten musi być spełniony. Niestety
nasze pragnienie leniuchowania doprowadza nas do paranoicznego strachu przed tym, że ktoś
złamie nasze warunki. A złamanie przyjętych warunków nieuchronnie prowadzi do błędów.
W wyniku tego zamiast pisać bloga, musimy siedzieć po nocach, analizować komunikaty o błędach,
szukać przyczyn niewłaściwego działania programu i ogólnie robić wiele rzeczy, na które nie
mamy ochoty. Gdybyśmy nie sprawdzali, czy zmienna tid jest mniejsza od N, i w pewnym
momencie pobrali zawartość pamięci, która do nas nie należy, to byśmy wpadli w tarapaty.
Mogłoby to nawet spowodować zakończenie działania jądra, ponieważ GPU mają wbudowane
wyrafinowane jednostki zarządzające pamięcią, które zamykają każdy proces, który by łamał
zasady korzystania z pamięci.

Jeśli w programie wystąpi tego rodzaju błąd, jedno z makr HANDLE_ERROR(), którymi szczodrze
sypiemy w całym kodzie, wykryje go i poinformuje Cię o tym. Należy pamiętać, że tak samo
jak w standardowym języku C, funkcje zwracają kody błędów nie bez powodu. Wiemy, że łatwo
ulec pokusie, aby zignorować pojawiający się kod błędu, ale chcielibyśmy zaoszczędzić Ci wielu
przykrych godzin, których sami nie zdołaliśmy uniknąć, i dlatego nalegamy, aby zawsze wery-
fikować wynik wszystkich działań, które mogą się nie udać. Jak to zwykle bywa, żaden z tych
błędów pewnie nie spowoduje natychmiastowego zamknięcia programu. Zamiast tego będą raczej
wywoływać najrozmaitsze nietypowe i nieprzyjemne efekty uboczne w dalszej perspektywie.

W tym momencie wiesz już, jak na GPU wykonać kod równolegle. Możliwe, że mówiono Ci,
iż jest to bardzo skomplikowane albo że trzeba znać się na programowaniu grafiki, aby tego
dokonać. Dotychczasowe przykłady stanowią jednak dowód na to, że dzięki językowi CUDA C
jest zupełnie inaczej. Ostatni program sumuje tylko dwa wektory zawierające po 10 elementów.
Jeśli chcesz zobaczyć równoległe wykonywanie kodu w pełnej skali, zmień w wierszu #define N
10 liczbę na 10000 albo 50000, tak aby utworzyć kilkadziesiąt tysięcy równoległych bloków wy-
konawczych. Pamiętaj tylko, że w każdym wymiarze maksymalna liczba bloków wynosi 65535.
Jest to ograniczenie sprzętowe, którego przekroczenie wywoła wiele różnych błędów w pro-
gramie. W następnym rozdziale nauczysz się pracować w tym wyznaczonym zakresie.

4.2.2. ZABAWNY PRZYKŁAD

Wcale nie twierdzimy, że dodawanie wektorów to nie jest świetna zabawa, ale teraz pokażemy
program, który zaspokoi wielbicieli bardziej wyszukanych efektów specjalnych.

Program ten będzie wyświetlał fragmenty zbioru Julii. Dla niewtajemniczonych wyjaśniamy, że
zbiór Julii to granica pewnej klasy funkcji w zbiorze liczb zespolonych. To chyba brzmi jeszcze
gorzej niż dodawanie wektorów czy mnożenie macierzy. Lecz dla prawie wszystkich wartości
parametrów tych funkcji granica ta tworzy fraktal, czyli jedną z najpiękniejszych i zarazem naj-
ciekawszych matematycznych osobliwości.

Obliczenia, jakie należy wykonać w celu wygenerowania takiego zbioru, są stosunkowo proste.
Wszystko sprowadza się do iteracyjnego rozwiązywania równania, którego parametrami są
punkty płaszczyzny zespolonej. Punkty, dla których ciąg rozwiązań równania dąży do nieskoń-
czoności, nie należą do zbioru. Natomiast punkty, dla których ciąg rozwiązań równania nie dąży
do nieskończoności, należą do zbioru.

Równanie, o które chodzi, pokazano na listingu 4.1. Jak widać, jest ono bardzo proste do obliczenia:

Listing 4.1.

CZZ 2
n1n ���

Aby więc obliczyć jedną iterację powyższego równania, należałoby podnieść do kwadratu bie-
żącą wartość i dodać stałą C. W ten sposób obliczyłoby się kolejną wartość równania.

ZBIÓR JULII NA CPU

Poniżej przedstawiamy kod źródłowy programu obliczającego i wizualizującego zbiór Julii.
Ponieważ jest on bardziej skomplikowany niż wszystkie poprzednie, podzieliliśmy go na części.
Dalej pokazany jest też ten kod w całości.

int main( void ) {
    CPUBitmap bitmap( DIM, DIM );
    unsigned char *ptr = bitmap.get_ptr();

    kernel( ptr );
    bitmap.display_and_exit();
}

Funkcja główna jest bardzo prosta. Tworzy przy użyciu funkcji bibliotecznej mapę bitową o od-
powiednim rozmiarze, a następnie do funkcji jądra przekazuje wskaźnik na tę mapę.

void kernel( unsigned char *ptr ){
    for (int y=0; y<DIM; y++) {
        for (int x=0; x<DIM; x++) {
            int offset = x + y * DIM;
            int juliaValue = julia( x, y );
            ptr[offset*4 + 0] = 255 * juliaValue;
            ptr[offset*4 + 1] = 0;
            ptr[offset*4 + 2] = 0;
            ptr[offset*4 + 3] = 255;
        }
    }
 }

Funkcja jądra po prostu przegląda iteracyjnie wszystkie punkty, które wyrenderujemy, i dla
każdego z nich wywołuje funkcję julia(), aby sprawdzić, czy należy on do zbioru, czy nie. Jeśli
dany punkt należy do zbioru, funkcja zwraca 1, jeśli nie — zwraca 0. W pierwszym przypadku
kolor punktu ustawiamy na czerwony, a w drugim na czarny. Wybór konkretnych kolorów nie
ma znaczenia, więc możesz ustawić swoje ulubione.

int julia( int x, int y ) {
    const float scale = 1.5;
    float jx = scale * (float)(DIM/2 - x)/(DIM/2);
    float jy = scale * (float)(DIM/2 - y)/(DIM/2);
    cuComplex c(-0.8, 0.156);
    cuComplex a(jx, jy);
    int i = 0;

    for (i=0; i<200; i++) {

        a = a * a + c;
        if (a.magnitude2() > 1000)
            return 0;
    }
    return 1;
}

Powyższa funkcja stanowi serce programu. Najpierw zamienia współrzędne piksela na współ-
rzędne na płaszczyźnie zespolonej. W celu wypośrodkowania tej płaszczyzny na obrazie stosu-
jemy przesunięcie o DIM/2. Następnie skalujemy każdą współrzędną o DIM/2, tak aby obraz
zajmował zakres od –1.0 do 1.0. Zatem dla dowolnego punktu (x,y) na płaszczyźnie zespolonej
otrzymujemy punkt ((DIM/2–x)/(DIM/2), (DIM/2–y)/(DIM/2)).

Aby umożliwić powiększanie i pomniejszanie obrazu, wprowadziliśmy współczynnik scale.
Aktualnie skala została ustawiona na sztywno na 1.5, ale można tę wartość dowolnie zmienić.
Bardziej ambitne osoby mogą nawet zdefiniować to ustawienie jako parametr wiersza poleceń.

Po obliczeniu współrzędnych punktu na płaszczyźnie zespolonej przechodzimy do sprawdzenia,
czy należy on do zbioru Julii. Pamiętamy, że aby to zrobić, trzeba obliczyć wartości rekurencyj-

nego równania CZZ 2
n1n ��� . Ponieważ C jest stałą liczbą zespoloną, której wartość można

dowolnie wybrać, ustawimy ją na –0.8 + 0.156i, gdyż wartość ta pozwala uzyskać bardzo cie-
kawy efekt. Warto skorzystać z tej możliwości, aby zobaczyć różne inne wersje zbioru Julii.

W prezentowanym programie obliczamy 200 iteracji funkcji. Po każdym powtórzeniu spraw-
dzamy, czy wartość bezwzględna wyniku nie przekracza pewnej ustalonej wartości (tu próg
ustawiliśmy na 1000). Jeśli tak, to przyjmujemy, że równanie dąży do nieskończoności, a więc
zwracamy 0, aby zaznaczyć, że dany punkt nie należy do zbioru. W przeciwnym razie, tzn. jeśli
po 200 iteracjach wartość nie przekracza 1000, przyjmujemy, że punkt należy do zbioru, i zwra-
camy 1 do wywołującego, czyli funkcji kernel().

Ponieważ wszystkie obliczenia są wykonywane na liczbach zespolonych, zdefiniowaliśmy ogólną
strukturę do ich przechowywania.

struct cuComplex {
    float   r;
    float   i;
    cuComplex( float a, float b ) : r(a), i(b)  {}
    float magnitude2( void ) { return r * r + i * i; }
    cuComplex operator*(const cuComplex& a) {
        return cuComplex(r*a.r - i*a.i, i*a.r + r*a.i);
    }
    cuComplex operator+(const cuComplex& a) {
        return cuComplex(r+a.r, i+a.i);
    }
};

Struktura ta zawiera dwie składowe reprezentujące liczbę zespoloną. Pierwsza z nich to liczba
zmiennoprzecinkowa pojedynczej precyzji o nazwie r reprezentująca część rzeczywistą, a dru-
ga to liczba zmiennoprzecinkowa pojedynczej precyzji o nazwie i, która reprezentuje część
urojoną. Dodatkowo w strukturze znajdują się definicje operatorów dodawania i mnożenia
liczb zespolonych (jeśli nie masz pojęcia o liczbach zespolonych, podstawowe wiadomości mo-
żesz szybko znaleźć w internecie). Ponadto w strukturze znajduje się definicja metody zwraca-
jącej wartość bezwzględną liczby zespolonej.

ZBIÓR JULII NA GPU

Implementacja dla GPU tradycyjnie jest bardzo podobna do implementacji dla CPU.

int main( void ) {
    CPUBitmap bitmap( DIM, DIM );
    unsigned char    *dev_bitmap;
    HANDLE_ERROR( cudaMalloc( (void**)&dev_bitmap,
                              bitmap.image_size() ) );
    dim3    grid(DIM,DIM);
    kernel<<<grid,1>>>( dev_bitmap );
    HANDLE_ERROR( cudaMemcpy( bitmap.get_ptr(),
                              dev_bitmap,
                              bitmap.image_size(),
                              cudaMemcpyDeviceToHost ) );
    bitmap.display_and_exit();
    cudaFree( dev_bitmap );
}

Mimo że ta wersja funkcji main() wygląda na bardziej skomplikowaną od poprzedniej, działa
dokładnie tak samo jak tamta. Najpierw przy użyciu standardowej funkcji bibliotecznej two-
rzymy mapę bitową o wymiarach DIM x DIM. Ponieważ obliczenia będą wykonywane na GPU,
dodatkowo zadeklarowaliśmy wskaźnik o nazwie dev_bitmap, który będzie wskazywał kopię
danych na urządzeniu. A do przechowywania tych danych potrzebna jest pamięć alokowana za
pomocą funkcji cudaMalloc().

Następnie (podobnie jak w wersji dla CPU) uruchamiamy funkcję kernel(), lecz tym razem
dodajemy do niej kwalifikator __global__, aby zaznaczyć, że ma ona zostać wykonana na GPU.
Tak jak poprzednio przekazujemy do niej utworzony wcześniej wskaźnik na miejsce w pamięci,
w którym mają być przechowywane dane. Jedyna różnica polega na tym, że teraz dane zamiast
w systemie hosta są przechowywane na GPU.

Największa różnica między tymi dwiema implementacjami polega na tym, że w wersji dla GPU
określona jest liczba bloków wykonawczych funkcji kernel(). Ponieważ obliczenia dla każdego
punktu można wykonywać niezależnie od pozostałych, utworzyliśmy po jednej kopii funkcji
dla każdego interesującego nas punktu. Wcześniej wspomnieliśmy, że w niektórych przypadkach
wygodniej jest używać indeksowania dwuwymiarowego. Jednym z nich jest właśnie obliczanie
wartości funkcji w dwuwymiarowej dziedzinie, takiej jak płaszczyzna zespolona. W związku
z tym poniższy wiersz zawiera definicję dwuwymiarowej siatki bloków:

dim3 grid(DIM,DIM);

Jeśli martwisz się, że zaczynasz zapominać podstawowe informacje, to pragniemy Cię uspoko-
ić, gdyż dim3 wcale nie jest standardowym typem języka C. W plikach nagłówkowych systemu
wykonawczego CUDA znajdują się definicje kilku typów pomocniczych reprezentujących
wielowymiarowe struktury. Typ dim3 reprezentuje krotkę trójwymiarową, jakiej użyjemy do
określenia liczby uruchomionych bloków. Ale dlaczego używamy trójwymiarowej wartości, skoro
wcześniej bardzo wyraźnie podkreślaliśmy, że utworzymy siatkę dwuwymiarową?

Zrobiliśmy to dlatego, że system wykonawczy CUDA oczekuje właśnie typu dim3. Mimo że
aktualnie trójwymiarowe siatki nie są obsługiwane, system wykonawczy CUDA wymaga
zmiennej typu dim3, w której ostatni element ma wartość 1. Jeśli do inicjacji tej zmiennej zosta-
ną podane tylko dwie wartości, tak jak w instrukcji dim3 grid(DIM,DIM), system automatycznie
wstawi w miejsce trzeciego wymiaru wartość 1, dzięki czemu program będzie działał poprawnie.
Możliwe, że w przyszłości NVIDIA doda obsługę także trójwymiarowych siatek, ale na razie
musimy grzecznie postępować z API wywoływania jądra, ponieważ w sporach między API
a programistą zawsze API jest górą.

Następnie zmienną grid typu dim3 przekazujemy do systemu wykonawczego CUDA za pomocą
poniższego wiersza kodu:

kernel<<<grid,1>>>( dev _ bitmap );

Ponieważ wyniki działania funkcji kernel() są zapisywane w pamięci urządzenia, trzeba je stamtąd
skopiować do hosta. Jak już wiemy, służy do tego funkcja cudaMemcpy() z ostatnim argumentem
wywołania cudaMemcpyDeviceToHost.

  HANDLE_ERROR( cudaMemcpy( bitmap.get_ptr(),
                            dev_bitmap,
                            bitmap.image_size(),
                            cudaMemcpyDeviceToHost ) );

Kolejna różnica między dwiema prezentowanymi wersjami dotyczy implementacji funkcji kernel():

__global__ void kernel( unsigned char *ptr ) {
    // Odwzorowanie z blockIdx na wspó�rz�dne piksela
    int x = blockIdx.x;
    int y = blockIdx.y;
    int offset = x + y * gridDim.x;

    // Obliczenie warto�ci dla tego punktu
    int juliaValue = julia( x, y );
    ptr[offset*4 + 0] = 255 * juliaValue;
    ptr[offset*4 + 1] = 0;

    ptr[offset*4 + 2] = 0;

    ptr[offset*4 + 3] = 255;
}

Po pierwsze, aby funkcja kernel() mogła być wywoływana z hosta, a wykonywana na urządzeniu,
musi zostać zadeklarowana jako __global__. W odróżnieniu od wersji dla CPU nie potrzebu-
jemy zagnieżdżonych pętli for() do generowania indeksów pikseli przekazywanych do funkcji
julia(). Podobnie jak było w przypadku dodawania wektorów, system wykonawczy CUDA
generuje je za nas w zmiennej blockIdx. Możemy skorzystać z tej możliwości dlatego, że
wymiary siatki bloków ustawiliśmy tak samo jak wymiary obrazu, dzięki czemu dla każdej pa-
ry liczb całkowitych (x,y) z przedziału od (0,0) do (DIM–1, DIM–1) otrzymujemy jeden blok.

Kolejna informacja, jakiej potrzebujemy, to pozycja w liniowym buforze wyjściowym ptr.
Obliczana jest ona przy użyciu innej standardowej zmiennej o nazwie gridDim. Jej wartość jest
stała we wszystkich blokach i reprezentuje wymiary siatki. W tym przypadku będzie to zawsze
wartość (DIM, DIM). Zatem mnożąc indeks wiersza przez szerokość siatki i dodając indeks ko-
lumny, otrzymamy indeks w ptr, należący do przedziału wartości od 0 do (DIM*DIM–1).

  int offset = x + y * gridDim.x;

Na koniec przeanalizujemy kod decydujący o tym, czy dany punkt należy do zbioru Julii. Jak
zwykle wygląda on bardzo podobnie jak implementacja dla CPU.

__device__ int julia( int x, int y ) {
    const float scale = 1.5;
    float jx = scale * (float)(DIM/2 - x)/(DIM/2);
    float jy = scale * (float)(DIM/2 - y)/(DIM/2);
    cuComplex c(-0.8, 0.156);
    cuComplex a(jx, jy);
    int i = 0;
    for (i=0; i<200; i++) {
        a = a * a + c;
        if (a.magnitude2() > 1000)
            return 0;
    }
    return 1;
}

W kodzie tym znajduje się definicja struktury cuComplex, która służy do reprezentacji liczb
zespolonych w postaci dwóch liczb zmiennoprzecinkowych pojedynczej precyzji. Ponadto
struktura ta zawiera definicje operatorów dodawania i mnożenia oraz funkcję zwracającą wartość
bezwzględną liczby zespolonej.

struct cuComplex {
    float   r;
    float   i;
    cuComplex( float a, float b ) : r(a), i(b)  {}

    __device__ float magnitude2( void ) {

        return r * r + i * i;
    }
    __device__ cuComplex operator*(const cuComplex& a) {
        return cuComplex(r*a.r - i*a.i, i*a.r + r*a.i);
    }
    __device__ cuComplex operator+(const cuComplex& a) {
        return cuComplex(r+a.r, i+a.i);
    }
};



PROGRAMOWANIE RÓWNOLEGŁE W JĘZYKU CUDA C

58

Zwróć uwagę, że w wersji CUDA C programu używane są takie same konstrukcje językowe jak
w wersji dla CPU. Jedyną różnicą jest użycie kwalifikatora __device__ oznaczającego, że dany
fragment kodu ma zostać wykonany na GPU. Należy pamiętać, że funkcje zadeklarowane jako
__device__ można wywoływać tylko z innych funkcji tego samego typu lub typu __global__.

Poniżej znajduje się w całości kod źródłowy opisanego programu.

#include "../common/book.h"
#include "../common/cpu_bitmap.h"
#define DIM 1000
struct cuComplex {
    float   r;
    float   i;
    cuComplex( float a, float b ) : r(a), i(b)  {}
    __device__ float magnitude2( void ) {
        return r * r + i * i;
    }
    __device__ cuComplex operator*(const cuComplex& a) {
        return cuComplex(r*a.r - i*a.i, i*a.r + r*a.i);
    }
    __device__ cuComplex operator+(const cuComplex& a) {
        return cuComplex(r+a.r, i+a.i);
    }
};
__device__ int julia( int x, int y ) {
    const float scale = 1.5;
    float jx = scale * (float)(DIM/2 - x)/(DIM/2);
    float jy = scale * (float)(DIM/2 - y)/(DIM/2);
    cuComplex c(-0.8, 0.156);
    cuComplex a(jx, jy);
    int i = 0;
    for (i=0; i<200; i++) {
        a = a * a + c;
        if (a.magnitude2() > 1000)
            return 0;
    }
    return 1;
}
__global__ void kernel( unsigned char *ptr ) {
    // Odwzorowanie z blockIdx na po�o�enie piksela
    int x = blockIdx.x;
    int y = blockIdx.y;
    int offset = x + y * gridDim.x;
    // Obliczenie warto�ci dla tego punktu
    int juliaValue = julia( x, y );
    ptr[offset*4 + 0] = 255 * juliaValue;
    ptr[offset*4 + 1] = 0;
    ptr[offset*4 + 2] = 0;
    ptr[offset*4 + 3] = 255;
}

int main( void ) {
    CPUBitmap bitmap( DIM, DIM );
    unsigned char    *dev_bitmap;
    HANDLE_ERROR( cudaMalloc( (void**)&dev_bitmap,
                               bitmap.image_size() ) );
    dim3    grid(DIM,DIM);
    kernel<<<grid,1>>>( dev_bitmap );
    HANDLE_ERROR( cudaMemcpy( bitmap.get_ptr(), dev_bitmap,
                              bitmap.image_size(),
                              cudaMemcpyDeviceToHost ) );
    bitmap.display_and_exit();
    HANDLE_ERROR( cudaFree( dev_bitmap ) );
}

Gdy uruchomisz ten program, zobaczysz wizualizację zbioru Julii. Jako dowód, że podrozdział ten
słusznie ma w tytule słowo „zabawny”, na rysunku 4.2 pokazany jest zrzut ekranu z tej aplikacji.

Rysunek 4.2. Zrzut ekranu z wersji GPU programu

4.3. Podsumowanie
Gratulacje! Potrafisz już pisać, kompilować i uruchamiać programy równoległe na procesorze
GPU. Koniecznie pochwal się znajomym. Jeśli nadal trwają oni w błędnym przekonaniu, że
programowanie GPU to egzotyczna i trudna do opanowania sztuka, to na pewno zrobisz na nich
piorunujące wrażenie. Jak udało Ci się tego dokonać, będzie naszym małym sekretem. A jeśli są
to ludzie, którym można bezpiecznie powierzyć tajemnice, powiedz im, żeby też kupili sobie tę
książkę.

W rozdziale tym pokazaliśmy, jak zmusić system wykonawczy CUDA do jednoczesnego wyko-
nywania wielu kopii jednego programu w tzw. blokach. Zbiór bloków uruchamianych na GPU
nazwaliśmy siatką. Zbiory bloków mogą być jedno- lub dwuwymiarowe. Korzystając ze zmiennej
blockIdx, można sprawdzić w każdej kopii funkcji jądra, który blok ją wykonuje. Analogicznie
dzięki wbudowanej zmiennej gridDim można sprawdzić rozmiar siatki. Obie te zmienne posłu-
żyły nam w programie do obliczenia indeksu danych do przetworzenia dla każdego z bloków.

zastosowania języka CUDA C, 22

obrazowanie medyczne, 23
ochrona środowiska, 24
symulacja dynamiki płynów, 24

zawartość pamięci stałej, 104
zawieszenie procesora, 83
zbiór bloków i wątków, 65
zbiór Julii na cpu, 52
zbiór Julii na gpu, 54
zbiór wątków, 98
zdarzenia, 99
zdarzenia CUDA, 100
zdefiniowany bufor pamięci, 74
zegar CPU, 18
zerowanie bufora w pamięci wspólnej, 154
zintegrowane GPU, 42
zintegrowany układ graficzny, 43
zmienna

blockDim, 64
blockIdx.x, 50
dragStartX, 131
dragStartY, 131
gridDim, 57, 60, 64
mutex, 206
offset, 118
tid, 51

zmienne zapisane w pamięci wspólnej, 73
zmniejszanie kolejek do pamięci, 156
znacznik cudaGraphicsMapFlagsNone, 126
znacznik cudaGraphicsMapFlagsWriteDiscard, 126
zrównoleglanie na poziomie danych, 157
zrównoleglanie na poziomie zadań, 157
zwalnianie buforów, 166
zwalnianie pamięci, 218

Ż
żądanie alokacji, 126






